

## David Ghan Con't

**Pets:** We have a calico cat called Toby. She is 4 years old.

**Sports/Hobbies:** I like to play the guitar and the viola. I like to play hockey early, early on Saturday mornings with a group of friends (when I can get up, that is).

**What your ideal weekend would be:** Waking up to a beautiful, calm, sunny day in Algonquin Park beside the lake after a good night's sleep, making a bacon and pancake breakfast, and then a day of canoeing, swimming and fishing with the family. More of the same on Sunday, but eggs and toast instead.

**Favourite Foods:** Greek salad, grilled meat, fruit pies. Generally whatever is on my plate at the time.

**If I could be anything at all (besides a SAS programmer), I would be....** A rock star.

**When I'm not programming in SAS, I like to .....** pretend I am a rock star. In fact I like to pretend I am a rock star WHILE I am programming in SAS.

**One thing every SAS programmer should know ....**

### Using Proc Template to Take Control of the Layout of Procedure Output

We begin by creating HTML output for Proc Means:

```
ods html file='means sol.html' style=sasweb ;
proc means data=ia.employees n min mean max p10 p25 p50 p75 p90 ;
  var salary;
  class JobCode;
  where jobcode contains 'FLT';
ods html close;
```

The output is very wide, due to all the statistics:

Analysis Variable : Salary Employee Salary										
Job Code	N Obs	N	Minimum	Mean	Maximum	10th Pctl	25th Pctl	50th Pctl	75th Pctl	90th Pctl
FLTAT1	170	170	16000.00	29594.12	45000.00	19000.00	23000.00	29000.00	37000.00	41000.00
FLTAT2	227	227	16000.00	30691.63	45000.00	19000.00	22000.00	31000.00	38000.00	43000.00
FLTAT3	354	354	16000.00	30861.58	45000.00	20000.00	23000.00	31000.00	39000.00	43000.00

Instead we will make the layout narrow and longer by stacking the statistics. This may permit all the information to fit within the line size limit of a printed document.

Below is the form of the output we will create:

Salary Statistics		
Job Code	Employees	Statistic
FLTAT1	170	min= 16000 mean= 29594 max= 45000 10th pctl= 19000 25th pctl= 23000 median= 29000 75th pctl= 37000 90th pctl= 41000
FLTAT2	227	min= 16000 mean= 30691 max= 45000 10th pctl= 19000 25th pctl= 22000 median= 31000 75th pctl= 38000 90th pctl= 43000
FLTAT3	354	min= 16000 mean= 30861 max= 45000 10th pctl= 20000 25th pctl= 23000 median= 31000 75th pctl= 39000 90th pctl= 43000

Note that 8 statistics have been stacked in a single column instead of being displayed in separate columns. Also, a picture format is used to display the name of each statistic to the left of the statistical value

The output above is still being produced by Proc Means by modifying the **table template** that SAS uses to determine the layout. Most SAS procedures use pre-defined table templates that determine how each output object will be displayed. These table template definitions can be altered to suit the desired look of the user.

For our objective the first task is to determine the name of the table template for Proc Means by using ODS Trace:

```
ods trace on / label;  
.  
  var salary;  
  class JobCode;  
  where jobcode contains 'FLT';  
run;  
ods trace off;
```

The SAS log:

```
Output Added:  
-----  
Name:      Summary  
Label:     Summary statistics  
Template:  base.summary  
Path:     Means.Summary
```

We can edit this template with the template procedure:

```
proc template;
  edit Base.Summary;
    column class n (min mean max p10
                   p25 p50 p75 p90)
                   ;
  header h;
  define h;
    text "Salary Statistics";
  end;
  edit min;
    header = "Statistic";
    format = minfmt.;
    generic;
    just=R;
  end;
  edit max;
    format = maxfmt.;
    generic;
    just=R;
  end;
  edit n;
    header= "Employees";
    format = comma8.;
    vjust=top;
    generic;
  end;
  edit mean;
    format = meanfmt.;
    generic;
    just=R;
  end;
  edit p10;
    just=R;
    format=p10fmt17.;
  end;
  edit p25;
    just=R;
    format=p25fmt17.;
  end;
  . . .
end;
run;
```



Column statement stacks all statistics from min to p90 vertically within one column by setting them within a set of brackets

Salary Statistics		
Job Code	Employees	Statistic
FLTAT1	170	min= 16000 mean= 29594 max= 45000 10th pctl= 19000 25th pctl= 23000 median= 29000 75th pctl= 37000 90th pctl= 41000
FLTAT2	227	min= 16000 mean= 30691 max= 45000 10th pctl= 19000 25th pctl= 22000 median= 31000

Note as well the use of picture formats to label each statistic within the table

The picture format was created using Proc Format:

```
proc format;
  picture minfmt    low-high=00009 (prefix='min= ');
  picture meanfmt   low-high=00009 (prefix='mean= ');
  picture maxfmt    low-high=00009 (prefix='max= ');
  picture p10fmt    low-high=00009 (prefix='10th pctl= ');
  picture p25fmt    low-high=00009 (prefix='25th pctl= ');
  picture p50fmt    low-high=00009 (prefix='median= ');
  picture p75fmt    low-high=00009 (prefix='75th pctl= ');
  picture p90fmt    low-high=00009 (prefix='90th pctl= ');
run;
```

Picture formats provide instructions about how to display numbers with characters before, embedded within, or after the numeric value.

The full program from start to finish is below:

```
ods path work.temptemp(update)
         sasuser.templat(update)
         sashelp.tmplmst(read);

ods listing close;
proc format;
  picture minfmt    low-high=00009 (prefix='min= ');
  picture meanfmt   low-high=00009 (prefix='mean= ');
  picture maxfmt    low-high=00009 (prefix='max= ');
  picture p10fmt    low-high=00009 (prefix='10th pctl= ');
  picture p25fmt    low-high=00009 (prefix='25th pctl= ');
  picture p50fmt    low-high=00009 (prefix='median= ');
  picture p75fmt    low-high=00009 (prefix='75th pctl= ');
  picture p90fmt    low-high=00009 (prefix='90th pctl= ');
run;
proc template;
  edit Base.Summary;
  column class n (min mean max p10 p25 p50 p75 p90) ;
  header h;
  define h;
    text "Salary Statistics";
  end;
  edit min;
  header = "Statistic";
  format = minfmt.;
  generic;
  just=R;
  end;
  edit max;
  format = maxfmt.;
  generic;
  just=R;
  end;
  edit n;
  header= "Employees";
  format = comma8.;
```

```

        vjust=top;
    generic;
end;
edit mean;
    format = meanfmt.;
    generic;
    just=R;
end;
edit p10;
    just=R;
    format=p10fmt17.;
end;
edit p25;
    just=R;
    format=p25fmt17.;
end;
edit p50;
    just=R;
    format=p50fmt17.;
end;
edit p75;
    just=R;
    format=p75fmt17.;
end;
edit p90;
    just=R;
    format=p90fmt17.;
end;
end;
run;

ods html file='means_sol.html' style=sasweb ;
title;
proc means data=ia.employees n min mean max p10 p25 p50 p75 p90 ;
    var salary;
    class JobCode;
    where jobcode contains 'FLT';
run;
ods html close;
ods listing;

```